


#### 4.5 BARORECEPTOR CONTROL OF THE CIRCULATORY SYSTEM

Pressures and flows in the uncontrolled circulation tend to be stabilized by the Frank-Starling mechanism (Noordergraaf-78); this may be shown by the partial recovery of cardiac output after a left ventricular infarction in model PF-0 of Section 4.3. However, the normal cardiovascular system has a more important stabilization mechanism as a result of feedback control acting through the central nervous system. This control depends on pressure signals that are converted to efferent nerve signals by the baroreceptors in the carotid arch. These nerve pathways are shown in Fig. 4.5.1, together with the afferent nerves, which carry signals from the baroreceptors to the central nervous system.

Studies of the baroreceptor control system (Katona-70 and 80;



**Figure 4.5.1.** The activity of the heart is affected by two sets of efferent nerves carrying signals from the CNS to the heart, the sympathetic and the parasympathetic (or vagus) nerves. These signals, in turn depend upon the afferent messages carried to the CNS from the baroreceptors.

Dick-68) have shown that the signals returned to the heart tend to reduce both heart rate and strength of contraction in response to increased pressure in the carotid arteries and to increase both if there is decreased pressure in the carotids. In addition, other signals lead to an increase in peripheral systemic resistance and decreased venous volume with decreasing carotid pressure, and vice versa.

It is interesting to add elementary negative feedback regulation, corresponding to the baroreceptor system, to the cardiovascular model PF-1; the new program will be called PF-1-REG. In this program, the pressure PA3M, corresponding approximately to the carotid pressure, is filtered by a simple lag with time constant TFIL = 3.2

$$PA3MF = \text{REALPL}(TFIL, PA3M, PA3MIC) \quad (4.5.1)$$

where DPA3MF = PA3MF - PREF is the error in pressure. Here pressures are in mm Hg, and both PREF and PA3MIC are set at 72 mm Hg for normal startup of PF-1-REG.

We now define

$$Z3 = KX * DPA3MF \quad (4.5.2)$$

where KX, a gain constant, is normally set to 0.2, and Z3 is a feedback quantity that may be positive or negative, according to whether the average carotid pressure is greater or less than PREF. We want a feedback quantity Y that is normally unity and bounded between the limits 0.1 and 1.9,

$$Y = \text{BOUND}(LLT, ULT, (1.0 + Z3)) \quad (4.5.3)$$

where LLT = 0.1, ULT = 1.9 .

The feedback signal Y is now used to control automatically the heart period TH, as well as the systolic period TS and the strength of contraction of both ventricles, according to the equations:

$$\begin{aligned} TH &= 0.2 + 0.6 * Y \\ TS &= 0.14 + 0.2 * TH \\ LS &= LSI / (0.5 * Y + 0.5) + FCNSW(TI, 0.0, 0.0, DLS) \\ RS &= RSI / (0.5 * Y + 0.5) + FCNSW(TI, 0.0, 0.0, DRS) \end{aligned} \quad (4.5.4)$$

Note that an increase in Y, which might result from an increase in carotid pressure, will tend to increase heart period (decrease heart rate) and also will decrease strengths of contraction of the ventricles. The result of these changes will tend to return arterial pressures, and thus cardiac output, back to their original levels.

The program PF-1-REG is shown next. Here the same scheme has

been used in INITIAL to find integrator initial conditions (chiefly for volumes) as in its predecessor, PF-1, although simpler methods could have been used to find initial conditions in this program.

```
PROGRAM PF-1-REG
INITIAL
  'Some constants and calculations of initial volumes'
  Constant QP1U=7.8, PP1EDM=7.2, CP1=.0002
  QP1IC= QP1U+ PP1EDM*CP1*1332.    $ 'Pulm. Art. 1'
  Constant QP2U=23.4, PP2EDM=7.0, CP2=.0004
  QP2IC= QP2U+ PP2EDM*CP2*1332.    $ 'Pulm. Art. 2'
  Constant QP3U=210.5, PP3EDM=6.6, CP3=.0027
  QP3IC= QP3U+ PP3EDM*CP3*1332.    $ 'Pulm. Art. 3'
  Constant QL1U=69., PL1EDM=4.45, CL1=.001
  QL1IC= QL1U+ PL1EDM*CL1*1332.    $ 'Pulm. Veins 1'
  Constant QL2U=69., PL2EDM=3.62, CL2=.001
  QL2IC= QL2U+ PL2EDM*CL2*1332.    $ 'Pulm. veins 2'
  Constant QLAU=814.5, PLAEDM=3.45, CLA=.01176
  QLAIC= QLAU+ PLAEDM*CLA*1332.    $ 'L. Atrium'
  Constant QLVU=10., PLVEDM=4.0, LD=45.
  QLVIC= QLVU+ PLVEDM*1332./LD     $ 'L. Ventricle'
  Constant QA1U=35.1, PA1EDM=64.3, CA1=.00018
  QA1IC= QA1U+ PA1EDM*CA1*1332.    $ 'Syst. Art. 1'
  Constant QA2U=.85., PA2EDM=64., CA2=.00023
  QA2IC= QA2U+ PA2EDM*CA2*1332.    $ 'Syst. Art. 2'
  Constant QA3U=710., PA3EDM=63., CA3=.00182
  QA3IC= QA3U+ PA3EDM*CA3*1332.    $ 'Syst. Art. 3'
  Constant QV1U=909., PV1EDM=13.5, CV1=.021
  QV1IC= QV1U+ PV1EDM*CV1*1332.    $ 'Syst. Veins 1'
  Constant QV2U=1948., PV2EDM=7.2, CV2=.045
  QV2IC= QV2U+ PV2EDM*CV2*1332.    $ 'Syst. Veins 2'
  Constant QRAU=1948., PRAEDM=6.64, CRA=.045
  QRAIC= QRAU+ PRAEDM*CRA*1332.    $ 'Rt. Atrium'
  Constant QRVU=10., PRVEDM=7.4, RD=68.
  QRVIC= QRVU+ PRVEDM*1332./RD     $ 'Rt. Ventricle'

  'Calc. of total blood vol. QT, total unstressed ...
  volume QU, and initial stressed volume, QS, at T=0.0'
  QT=QP1IC+ QP2IC+ QP3IC+QL1IC+QL2IC+ QLAIC+ QLVIC...
  +QA1IC+ QA2IC+ QA3IC+ QV1IC+ QV2IC+ QRAIC+ QRVIC
  QU=QP1U+ QP2U+ QP3U+ QL1U+ QL2U+ QLAU+ QLVU+ QA1U...
  +QA2U+ QA3U+ QV1U+ QV2U+ QRAU+ QRVU
  QS= QT- QU
```

END \$'of initial'

DYNAMIC

Constant TF = 8.

```

TERMT(T .GE. TF)
Cinterval CINT=.02
DERIVATIVE
  Algorithm IALG = 4      $ '2nd order RK'
  Maxinterval MAXT = .002 $ Nsteps NSTP = 1

Constant THI=800., LSI=2500.,RSI=350.,DLS=0.,DRS=0.
  'DLS and DRS are values of changes in LS, RS at THI'
  'These changes will be negative for an infarct'
LS= LSI/(0.5*Y + 0.5) + FCNSW(TI,0.,0.,DLS)
RS= RSI/(0.5*Y + 0.5) + FCNSW(TI,0.,0.,DRS)
TI=T - THI              $ 'Infarct at THI'

Constant PI=3.1416,KB=1.,SV1=.9,SV2=.25
Constant TFIL=3.2,PA3MIC=72., LLT=0.1,ULT=1.9,PREF=72.
PA3M=PA3/1332.
PA3MF= REALPL(TFIL,PA3M,PA3MIC)      $ 'Filter Carotid Press.'
DPA3MF=PA3MF -PREF
Z3 = KX * DPA3MF
Constant KX = .02                    $ 'Use KX=0.0 for no Reg.'
Y = BOUND(LLT,ULT,(1.0 + Z3))        $ 'Y appears above in LS, RS'
TH = 0.2 + 0.6*Y                    $ TS = 0.14 + 0.2 *TH

LOGICAL XX
X=T-ZOH(T,0.0,0.0,TH)
XX=X .LE. TS
STW=RSW(XX,X,0.0)
SSW=SV1*SIN(PI*STW/TS)-SV2*SIN(2.*PI*STW/TS)
ACTV=KB* BOUND(0.0,1.0,SSW)         $ 'Ventr. Pumping Activ.'

  'Note, TH appears in Eqn. for X, TS in XX and SSW'
  'Pressure-Flow Equations start here'
Constant RPW1=10.,LP1=1.0,FP1IC=0.0, . . .
  KP1=1.,RP1=10.                    $ 'Pul. Art. 1'
PP1= (QP1-QP1U)/CP1 +KP1 *RPW1*(FRV-FP1)
FP1= INTEG((PP1-PP2-RP1*FP1)/LP1,FP1IC)
QP1= INTEG(FRV-FP1,QP1IC)           $ 'CP1 AND QP1IC in INITIAL'
Constant RP2=40.                    $ 'Pul. Art. 2'
PP2= (QP2-QP2U)/CP2
FP2= (PP2-PP3)/RP2
QP2= INTEG(FP1-FP2,QP2IC)
Constant RP3=80.                    $ 'Pul. Art. 3'
PP3= (QP3-QP3U)/CP3
FP3= (PP3-PL1)/RP3
QP3= INTEG(FP2-FP3,QP3IC)
Constant RL1=30.                    $ 'Pul. Vein 1'
PL1= (QL1-QL1U)/CL1

```

```

FL1= (PL1-PL2)/RL1
QL1= INTEG(FP3-FL1,QL1IC)
Constant RL2=10.,LL2=1.0,FL2IC=33. $ 'Pul. Vein 2'
PL2= (QL2-QL2U)/CL2
FL2= INTEG((PL2-PLA-RL2*FL2)/LL2,FL2IC)
QL2= INTEG(FL1-PL2,QL2IC)
Constant RLA=5.,LLA=1.0,FLAIC=0.0 $ 'Left Atrium'
PLA= (QLA-QLAU)/CLA
FLA= LIMINT((PLA-PLV-RLA*FLA)/LLA,FLAIC,0.0,1.E4)
QLA= INTEG(FL2-FLA,QLAIC)
Constant RLV=5.,LLV=1.,FLVIC=0.0 $ 'Left Ventr.'
SLV= D*(1.-ACTV) + LS*ACTV
  'LD given in INITIAL, LS in 4-th line of DERIVATIVE'
PLV= (QLV-QLVU)*SLV
FLV= LIMINT((PLV-PA1-RLV*FLV)/LLV,FLVIC,0.,1.E5)
QLV= INTEG(FLA-PLV,QLVIC)
Constant FA1IC=4.6,RA1=10.,LA1=1.0,FI=0.$ 'Aorta 1'
  'Note:RPW1, KP1 given above PP1'
PA1= (QA1-QA1U)/CA1 +KP1*RPW1*(FLV-PA1)
FA1= INTEG((PA1-PA2-RA1*FA1)/LA1,FA1IC)
QA1= INTEG(FLV-FA1+FI,QA1IC)
Constant RA2=160.                  $ 'Aorta 2'
PA2= (QA2-QA2U)/CA2
FA2= (PA2-PA3)/RA2
QA2= INTEG(FA1-FA2,QA2IC)
Constant RA3=1000.                 $ 'System. Art.'
PA3= (QA3-QA3U)/CA3
FA3= (PA3-PV1)/RA3
QA3= INTEG(FA2-FA3,QA3IC)
Constant RV1=90.                   $ 'System. Veins 1'
PV1= (QV1-QV1U)/CV1
FV1= (PV1-PV2)/RV1
QV1= INTEG(FA3-FV1,QV1IC)
Constant RV2=10.,LV2=1.,FV2IC=95. $ 'System. Veins 2'
PV2= (QV2-QV2U)/CV2
FV2= INTEG((PV2-PRA-RV2*FV2)/LV2,FV2IC)
FV2F=REALPL(3.2,FV2,FV2FIC)        $ 'Use for Cardiac Output'
Constant FV2FIC=78.
Constant GBS=0.0,TB=3.0
GB = FCNSW(T-TB,0.,0.,GBS)
FB = FV2*GBS                      $ 'Venous bleeding at TB if GBS > 0.0'
QV2= INTEG(FV1-FV2-FB,QV2IC)
Constant RRA=5.,LRA=1.0,FRAIC=0. $ 'Rt. Atrium'
PRA= (QRA-QRAU)/CRA
FRA= LIMINT((PRA-PRV-RRA*FRA)/LRA, FRAIC, 0.,1.E4)
QRA= INTEG (FV2-FRA,QRAIC)
Constant RRV=5.,LRV=1.,FRVIC=6.0 $ 'Rt. Ventr.'
SRV= RD*(1.-ACTV) + RS*ACTV

```

```

PRV= (QRV-QRVU)*SRV
FRV= LIMINT((PRV-PP1-RRV*FRV)/LRV,FRVIC,0.,1.E5)
QRV= INTEG(FRA-FRV,QRVIC)
END $'of Deriv.'

'Calculate Output Pressures in mmHg'
PP1M=PP1/1332. $ PP2M=PP2/1332. $ PP3M=PP3/1332.
PL1M=PL1/1332. $ PL2M=PL2/1332. $ PLAM=PLA/1332.
PLVM=PLV/1332. $ PA1M=PA1/1332. $ PA2M=PA2/1332.
PA3M=PA3/1332. $ PV1M=PV1/1332. $ PV2M=PV2/1332.
PRAM=PRA/1332. $ PRVM=PRV/1332.
QTOT=QP1+QP2+QP3+QL1+QL2+QLA+QLV+QA1+QA2...
+QA3+QV1+QV2+QRA+QRV $'Total blood'
QSTOT=QTOT-QU $'Total blood,
stressed'
END $ 'of Dynamic'
END $ 'of Program'

```

Some open-loop outputs obtained with this model are shown in Fig. 4.5.2; here feedback was removed by setting  $KX = 0.0$ , so that  $Y$  remains at unity, giving normal steady-state operation. At  $T = 11.0$  the left ventricle stiffness was reduced by 650.0, giving responses to a sudden infarction. The filtered carotid pressure PA3MF for this open-loop case is shown.

Closed-loop response, with  $KX$  set at 0.2, appears in Fig. 4.5.3; note that at the time of infarction (of the left ventricle, as in the open-loop responses of Fig. 4.5.2), the arrows in (4.5.5) show the expected sequence of responses,

$$\begin{aligned}
 &SL \downarrow, PLVM \downarrow, QLV(\text{peak-to-peak}) \downarrow, PA3M \downarrow, \\
 &Y \downarrow, TH \downarrow, SL \uparrow
 \end{aligned}
 \quad (4.5.5)$$

The resulting increased heart rate and strength of both ventricles tend to counter the effect of the original system parameter disturbance.

The effectiveness of the control system may also be studied by comparing open- and closed-loop plots of system response to other sudden disturbances, such as valve failure, opening of an anastomosis, hemorrhage, or opening (or closing) of a ventricular or atrial septal defect. Note that in PF-1-REG the arterial infusion FI is simpler than in PF-1, in that it can only start at  $T = 0.0$ ; however, venous bleeding is now possible, starting at time TB, as determined by the choice of bleeding conductance GBS, according to

$$FB = PV2 * GBS \quad (4.5.6)$$

In case of bleeding it may be interesting to determine changes in blood volume using the commands at the end of Dynamic (See Prob.4.7e).





Figure 4.5.2. Open loop response of the system PF-1-REG with no baroreceptor feedback ( $Y$  held at 1.0), showing normal operation followed by an infarct at  $T = 11$ .

(Note that plots begin at  $T = 8.0$ ).


(a) Left ventricular volume.

(b) Left ventricular pressure, PLVM, and carotid pressure, PA3MF (after filtering).



**Figure 4.5.3.** Closed-loop response of the cardiovascular system (program PF-1-REG, with  $KX = 0.2$ ), to a left ventricle infarct at  $T = 11.0$  sec.  
 (a) Feedback signal,  $Y$ , and left ventricular volume  $QLV$ .  
 (b) Heart period  $TH$  and right ventricular volume  $QRV$ .  
 (c) Ventricle muscle stiffnesses  $LS$  and  $RS$ .  
 (Note that  $LS$  drops suddenly at  $T = 11.$ , but after one oscillation,  $LS$  partially recovers, and  $RS$  shows a slight increase.)  
 (d) The feedback  $Y$  and heart period  $TH$  respond to the infarct with lower values and some oscillation, but final  $TH$  and  $Y$  tend to recover after  $T = 11$ .

A more detailed and realistic model of the baroreceptor control system (Tham-88) will include a better representation of the responses of the central nervous system, as well as feedback effects which change the flow resistances and compliances. This model is based on one introduced by Katona, and used by Dick and Tham (Katona-80, Dick-68, Tham-88). It is



**Figure 4.5.4.** Baroreceptor Feedback Control Loop

shown in block form in Fig. 4.5.4, depicting the neural feedback system for control of blood pressure. Note the "push-pull" action provided by the opposing sympathetic and vagus channels. The vertical arrows on variables indicate the results if the loop is opened in the feedback line and  $P_{sys}$  is applied at the lower left; the "returned"  $P_{sys}$  at the right of the diagram is of the opposite sign to that first applied, indicating that the system would serve to correct the pressure  $P_{sys}$  toward  $P_{ref}$  if the loop were closed.